
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 8.2.2.6.2 which is part of the LTE/SAE test suite in the ‘iwd-TTCN3-B2013-03_D15wk50’ ATS delivery.

The test case can be demonstrated to run with one LTE UE (see section 5) on multiple test platforms. Execution logs are provided as evidence.

3. Verification Test Summary

Test Case:
TC_8_2_2_6_2
Test Group:
LTE_A\8_2\ RRC_ConnReconfig_A.ttcn
ATS Version:
iwd-TTCN3-B2013-03_D15wk50
System Simulator used:
R&S CMW500
UE used:
MTK Jade Handset
Verification Status:
PASS


4. Corrections required for TC 8.2.2.6.2
Change 1 – Correction to function ‘f_TC_8_2_2_6_2_EUTRA’ 
	Function name
	f_TC_8_2_2_6_2_EUTRA()

	Reason for change
	1. Variables v_SR_PUCCH_ResourceIndex_Cell1 and v_SR_ConfigIndex_Cell1 are being initialized before the initialization of the global cell array.

2. According to TS 36.523, Table 8.2.2.6.2.3.2-1, at T1, cell-specific RS EPRE of cell 2 should be -79 dBm. However, this would cause issues at Step 7 since the power level of Cell 2 (-79 dBm) is stronger than Cell 1 (-85 dBm) leading to a Fatal Error whilst setting the attenuation in function fl_EUTRA_ChangeCellAttenuation as variable v_AttenuationValue < 0.
3. Variable v_Ncc2 is being initialized to 2. However, According to TS 36.508 Table 4.6.5-1 and Table 4.6.4, the default configuration for Security Config is used in the RRC Connection Reconfig message sent at Step 9 of the Prose. Here, the NextHopChainingCount is always assigned to 0 and the KeyChangeIndicator is set to FALSE. Hence, the variable v_Ncc2 should be renamed to v_Ncc0 and be assigned to 0.
4. Timer variables v_T304 and v_T340 are being initialized before the initialization of the global cell array. This causes the TC to fail due to Invalid values in the cell array
5. The timer T304 is being classed as an rrcTimer and a max tolerance function is being invoked. The duration of this timer 1000ms and with max tolerance the duration is increased to 1140ms. However, on the UE side, T304 would run for 1000ms. Due to the additional 140ms for the expiry of T304 on the SS side, Cell 2 would still be active and would be broadcasting System information messages. This causes the UE to reselect back to Cell 2 leading to TC failure
6. The function f_EUTRA_RRCConnectionReconfiguration_MeasConfig already has provision to receive the RRC Connection Reconfiguration Complete message. Hence, an additional receive statement is not needed.

7. The function f_EUTRA_508RRC_IntraLTE_HO_InterCell is a success scenario function which completes the HO procedure. However, in this TC, a HO failure is required. Hence, it would only be necessary to send the RRC Connection Reconfiguration message on Cell 1 to order the UE to perform intra frequency HO to Cell 2 accompanied with the re-setting of SRBs and DRBs on Cell 1
8. The second parameter to function f_Calculate_ShortMAC should be Cell 1 rather than Cell 2 to correctly calculate Short MAC-I
9. Security should be re-activated on Cell 1

	Summary of change
	1.  Initialised variables v_SR_PUCCH_ResourceIndex_Cell1 and v_SR_ConfigIndex_Cell1after the call to function  f_UTRA_Init

2. Invoked function f_EUTRA_CellInfo_InitMaxReferencePower to set the power level of Cell 2 to -79dBm

3. Changed name of variable v_Ncc2 to v_Ncc0 and initilaised it to 0.

4. Initialised the timer variables v_T304 and v_T340 after the call to function  f_UTRA_Init
5. Invoked the function f_EUTRA_SetTimerToleranceMin to ensure that Cell 2 is swiched off in time so that the UE does not re-select back to it.
6. Commented out the line which receives the RRCConnectionReconfiguration Complete message at Step 2. This change also applies at Step 17.

7. Created a variable v_RRCConnectionReconfiguration to form the HO message and send it. Also, the SRBs and DRBs on Cell 1 have been re-set.

8. Changed the second parameter of the function f_Calculate_ShortMAC to be Cell 1

9. Invoked function f_EUTRA_SS_AS_ActivateSecurity_HO_Reest to re-activate security on Cell 1 as well as re-establish DRBs

	TTCN module
	\LTE_A\8_2\ RRC_ConnReconfig_A.ttcn

	MCC160 Comment
	


Before change

...

function f_TC_8_2_2_6_2_EUTRA() runs on EUTRA_PTC

  {/*  @sic R5-155785 sic@ Test case 8.2.2.6.2 RRC connection reconfiguration/ UE Assistance Information/power preference indication release on connection re-establishment */

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var template (value) CellPowerList_Type v_CellPowerList_AtT0, v_CellPowerList_AtT1, v_CellPowerList_AtT2;

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_CellInfo_Cell1;

    var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo_Cell1;

    var SR_PUCCH_ResourceIndex_Type v_SR_PUCCH_ResourceIndex_Cell1 := f_EUTRA_CellInfo_GetSR_PUCCH_ResourceIndex (eutra_Cell1);

    var SR_ConfigIndex_Type v_SR_ConfigIndex_Cell1 := f_EUTRA_CellInfo_GetSR_ConfigIndex (eutra_Cell1);

    var NextHopChainingCount v_Ncc2 := 2;

    var ShortMAC_I v_ShortMAC_I;

    var PhysCellId v_PhysCellId_Cell1, v_PhysCellId_Cell2;

    var Frequency_IE_Type v_Frequency_IE_Cell2;

    var ChannelBandwidthDependency_Type v_ChBandDependency_Cell2;

    //Table 8.2.2.6.2.3.3-1: OtherConfig-r9

    var PowerPrefIndicationConfig_r11.setup.powerPrefIndicationTimer_r11 v_PowerPrefIndicationTimer := s600;

    var float v_T600 := 600.0;

    var float v_T304 := f_EUTRA_SetTimerToleranceMax(eutra_Cell2, rrcTimer, 1.0); //According to 36.508 cl. 4.6.5 Table 4.6.5-1

    var float v_T340 := f_EUTRA_SetTimerToleranceMax ( eutra_Cell1, rrcTimer, v_T600 ); //s600

    timer t_T304;

    timer t_T340;

    v_CellPowerList_AtT0 := {

      cs_CellPower (eutra_Cell1, tsc_ServingCellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE)

    };

    v_CellPowerList_AtT1 := {

      cs_CellPower (eutra_Cell2, -79)

    };

    v_CellPowerList_AtT2 := {

      cs_CellPower (eutra_Cell2, tsc_NonSuitableOffCellRS_EPRE)

    };

    f_EUTRA_Init (c2);

    v_PhysCellId_Cell1 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1);

    v_PhysCellId_Cell2 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell2);

    v_Frequency_IE_Cell2 := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell2);

    v_ChBandDependency_Cell2 := f_EUTRA_BandDependentParam(v_Frequency_IE_Cell2.DL_ChBandwidth, v_Frequency_IE_Cell2.UL_ChBandwidth);

    v_EUTRA_FDD_TDD_CellInfo_Cell1 := f_EUTRA_CellInfo_GetFDD_TDD_Info (eutra_Cell1);

    v_AntennaInfo_Cell1 := f_EUTRA_CellInfo_GetAntennaInfoDedicated (eutra_Cell1);

    //Create and configure all cells

    f_EUTRA_CellConfig_Def (eutra_Cell1);

    f_EUTRA_CellConfig_Def (eutra_Cell2);

    //Bring UE to initial state

    f_EUTRA_Preamble (eutra_Cell1, STATE2_IDLEUPDATE);

    f_EUTRA_RbEst_Def (eutra_Cell1);

    //Set initial cell power levels

    f_EUTRA_SetCellPowerList (v_CellPowerList_AtT0);

    f_EUTRA_TestBody_Set (true);

    //@siclog "Step 1" siclog@

    //The SS transmits an RRCConnectionReconfiguration message to setup intra frequency measurement on Cell 1 & to

    //configure power preference indication.

    f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell1,

                                                    cs_RRCConnectionReconfiguration_8_2_2_6_2_Step1(v_RRC_TI,

                                                                                                  v_Frequency_IE_Cell2.UL_DL_Earfcn.dl_CarrierFreq,

                                                                                                  v_ChBandDependency_Cell2.AllowedMeasBandwidth,

                                                                                                  v_PowerPrefIndicationTimer));

    //@siclog "Step 2" siclog@

    //The UE transmits an RRCConnectionReconfigurationComplete message on Cell 1

    SRB.receive( car_SRB1_RrcPdu_IND( eutra_Cell1, cr_508_RRCConnectionReconfigurationComplete(v_RRC_TI)));

    //@siclog "Step 3" siclog@

    //Cause the UE to send UEAssistanceInformation message indicating "lowPowerConsumption". (Note1).

    f_UT_PowerPreferenceIndication(UT);

    //@siclog "Step 4" siclog@

    //UE transmits UEAssistanceInformation message with powerPrefIndication IE set to lowPowerConsumption

    SRB.receive (car_SRB1_RrcPdu_IND (eutra_Cell1, cr_508_UeAssistanceInformation (lowPowerConsumption)));

    //@siclog "Step 5" siclog@

    //Cause the UE to send UEAssistanceInformation message indicating "normal" (Note2).

    f_UT_PowerPreferenceIndication(UT,'0'B);

    //@siclog "Step 6" siclog@

    //UE transmits UEAssistanceInformation message with powerPrefIndication IE set to normal

    //and starts timer T340 with the timer value set to the powerPrefIndicationTimer.

    SRB.receive (car_SRB1_RrcPdu_IND (eutra_Cell1, cr_508_UeAssistanceInformation (normal)));

    t_T340.start(v_T340);

    //@siclog "Step 7" siclog@

    //The SS changes Cell 1 and Cell 2 parameters according to the row "T1" in table 8.2.2.6.2.3.2-1

    f_EUTRA_SetCellPowerList (v_CellPowerList_AtT1);

    //@siclog "Step 8" siclog@

    //The UE transmits a MeasurementReport message on Cell 1 to report event A3 with the measured RSRP,

    //RSRQ value for Cell 2

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1,

                                    cr_MeasurementReport_8_2_2_6_2_Step8(v_PhysCellId_Cell2)));

    //Configure SS to not send response on RACH preamble reception on Cell 2

    f_EUTRA_SS_ConfigRachProcedure ( eutra_Cell2,

                                     omit,

                                     cs_RachProcedureConfig_NoResponse ( f_EUTRA_CellInfo_GetDL_ChBandwidth ( eutra_Cell2 ) ) );

    t_T304.start(v_T304);

    //@siclog "Step 9" siclog@

    //The SS transmits an RRCConnectionReconfiguration message on Cell 1 to order the UE

    //to perform intra frequency handover to Cell 2

    f_EUTRA_508RRC_IntraLTE_HO_InterCell(eutra_Cell1, eutra_Cell2);

    /*

    EXCEPTION: The steps 1 and 2 below are repeated for the duration of T304.

     1- The UE attempts to perform the intra frequency handover using MAC Random Access Preamble on Cell 2.

     2- The SS does not respond.

    */

    t_T304.timeout;

    //Configure Cell 1 for normal PRACH operation

    f_EUTRA_SS_ConfigRachProcedure_Def(eutra_Cell2, tsc_C_RNTI_Def);

    //@siclog "Step 10" siclog@

    //The SS changes Cell 1 and Cell 2 parameters according to the row "T2" in table 8.2.2.6.2.3.2-1

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT2);

    //Calculate ShortMAC-I to be received in RRCConnectionReestablishment Request from UE on Step11

    v_ShortMAC_I := f_Calculate_ShortMAC ( eutra_Cell1, eutra_Cell2 );

    //@siclog "Step 11" siclog@

    //The UE releases the configured powerPrefIndicationConfig, stops timer T340

    //and transmit an RRCConnectionReestablishmentRequest message on Cell 1

    t_T340.stop;

    SRB.receive (car_SRB0_RrcPdu_IND (eutra_Cell1,

                                      cr_508_RRCConnectionReestablishmentRequest (tsc_C_RNTI_Def,

                                                                                  v_PhysCellId_Cell1,

                                                                                  cr_ReestablishmentCause_HandoverFailure,

                                                                                  v_ShortMAC_I)));

    //@siclog "Step 12" siclog@

    //The SS transmits an RRCConnectionReestablishment message to resume SRB1 operation and re-activate security on Cell 1

    f_EUTRA_RRC_RRCConnectionReestablishment_Def (eutra_Cell1, v_RRC_TI, v_Ncc2);

    //@siclog "Step 13" siclog@

    //The UE transmits an RRCConnectionReestablishmentComplete message on Cell 1

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_508_RRCConnectionReestablishmentComplete(v_RRC_TI)));

    //@siclog "Step 14" siclog@

    //Cause the UE to send UEAssistanceInformation message indicating "lowPowerConsumption". (Note1).

    f_UT_PowerPreferenceIndication(UT, -, true);

    //@siclog "Step 15" siclog@

    //Check: Does the UE transmit UEAssistanceInformation message within the next 20s? (Note 3)

    fl_WaitForUeAssistanceInformation(-, "Test Case 8.2.2.6.2 Step 15");

    //Table 8.2.2.6.2.3.3-12: OtherConfig-r9

    v_PowerPrefIndicationTimer := s10;

    //@siclog "Step 16" siclog@

    //The SS transmits an RRCConnectionReconfiguration message to resume existing radio bearer on Cell 1

    //and to configure power preference indication

    f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell1,

                                                    cs_RRCConnReconfig_Resume_SRB2_DRBs_ConfigurePowerPreferenceIndication(v_RRC_TI,

                                                                                        v_EUTRA_FDD_TDD_CellInfo_Cell1.cqi_ReportConfig,

                                                                                        v_EUTRA_FDD_TDD_CellInfo_Cell1.soundingRS_UL_Config,

                                                                                        v_AntennaInfo_Cell1,

                                                                                        v_SR_PUCCH_ResourceIndex_Cell1,

                                                                                        v_SR_ConfigIndex_Cell1,

                                                                                        omit,

                                                                                        omit,

                                                                                        v_PowerPrefIndicationTimer));

    //@siclog "Step 17" siclog@

    //The UE transmits an RRCConnectionReconfigurationComplete message on Cell 1

    SRB.receive( car_SRB1_RrcPdu_IND( eutra_Cell1, cr_508_RRCConnectionReconfigurationComplete(v_RRC_TI)));

    //@siclog "Step 18" siclog@

    //Cause the UE to send UEAssistanceInformation message indicating "lowPowerConsumption". (Note1).

    f_UT_PowerPreferenceIndication(UT, -, true);

    //@siclog "Step 19" siclog@

    //Check: Does the UE transmit UEAssistanceInformation message with powerPrefIndication IE

    //set to lowPowerConsumption? (Note 4)

    v_T340 := f_EUTRA_SetTimerToleranceMax ( eutra_Cell1, rrcTimer, 10.0 ); //s10

    t_T340.start(v_T340);

    alt{

        [] SRB.receive (car_SRB1_RrcPdu_IND (eutra_Cell1, cr_508_UeAssistanceInformation (lowPowerConsumption)))

        {

            t_T340.stop;

            f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 8.2.2.6.2 Step 19");

        }

        [] t_T340.timeout

        {

            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 8.2.2.6.2 Step 19");

        }

    }

    f_EUTRA_TestBody_Set (false);

    //Switch/power off UE

    f_EUTRA_Postamble (eutra_Cell1, E2_CONNECTED);

  }//end of TC8.2.2.6.2
...

After change

...

  import from EUTRA_SecurityFunctions all; //WA#
  function f_TC_8_2_2_6_2_EUTRA() runs on EUTRA_PTC

  {/* @desc Test case 8.2.2.6.2 RRC connection reconfiguration/ UE Assistance Information/power preference indication release on connection re-establishment */

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var template (value) CellPowerList_Type v_CellPowerList_AtT0, v_CellPowerList_AtT1, v_CellPowerList_AtT2;

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_CellInfo_Cell1;

    var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo_Cell1;

    var SR_PUCCH_ResourceIndex_Type v_SR_PUCCH_ResourceIndex_Cell1; //WA#
    var SR_ConfigIndex_Type v_SR_ConfigIndex_Cell1; //WA#
    var NextHopChainingCount v_Ncc0 := 0; //WA#
    var ShortMAC_I v_ShortMAC_I;

    var EUTRA_SecurityParams_Type v_Auth_Params; //WA#
    var PhysCellId v_PhysCellId_Cell1, v_PhysCellId_Cell2;
    var Frequency_IE_Type v_Frequency_IE_Cell2;
    var ChannelBandwidthDependency_Type v_ChBandDependency_Cell2;
    //Table 8.2.2.6.2.3.3-1: OtherConfig-r9

    var PowerPrefIndicationConfig_r11.setup.powerPrefIndicationTimer_r11 v_PowerPrefIndicationTimer := s600;

    var template (value) DL_DCCH_Message v_RRCConnectionReconfiguration; //WA#
    var float v_T600 := 600.0;

    var float v_T304; //WA#
    var float v_T340; //WA#
    timer t_T304;

    timer t_T340;

    v_CellPowerList_AtT0 := {

      cs_CellPower (eutra_Cell1, tsc_ServingCellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE)
    };

    v_CellPowerList_AtT1 := {

      cs_CellPower (eutra_Cell2, -79)
    };

    v_CellPowerList_AtT2 := {

      cs_CellPower (eutra_Cell2, tsc_NonSuitableOffCellRS_EPRE)
    };

    f_EUTRA_Init (c2);

    //Set cell powel level for cell 1 and 2 to be used at creation

    f_EUTRA_CellInfo_InitMaxReferencePower ( eutra_Cell2, -79 ); //WA#
    //WA#
    v_SR_PUCCH_ResourceIndex_Cell1 := f_EUTRA_CellInfo_GetSR_PUCCH_ResourceIndex (eutra_Cell1);

    v_SR_ConfigIndex_Cell1 := f_EUTRA_CellInfo_GetSR_ConfigIndex (eutra_Cell1);
    v_T304:= f_EUTRA_SetTimerToleranceMin(eutra_Cell2, rrcTimer, 1.0); //According to 36.508 cl. 4.6.5 Table 4.6.5-1 WA#
    v_T340:= f_EUTRA_SetTimerToleranceMax ( eutra_Cell1, nonProtocolTimer, v_T600 ); //s600 WA#
    v_PhysCellId_Cell1 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1);

    v_PhysCellId_Cell2 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell2);
    v_Frequency_IE_Cell2 := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell2);
    v_ChBandDependency_Cell2 := f_EUTRA_BandDependentParam(v_Frequency_IE_Cell2.DL_ChBandwidth, v_Frequency_IE_Cell2.UL_ChBandwidth);
    v_EUTRA_FDD_TDD_CellInfo_Cell1 := f_EUTRA_CellInfo_GetFDD_TDD_Info (eutra_Cell1);

    v_AntennaInfo_Cell1 := f_EUTRA_CellInfo_GetAntennaInfoDedicated (eutra_Cell1);

    //Create and configure all cells

    f_EUTRA_CellConfig_Def (eutra_Cell1);

    f_EUTRA_CellConfig_Def (eutra_Cell2);
    //Bring UE to initial state

    f_EUTRA_Preamble (eutra_Cell1, STATE2_IDLEUPDATE);

    f_EUTRA_RbEst_Def (eutra_Cell1);

    //Set initial cell power levels

    f_EUTRA_SetCellPowerList (v_CellPowerList_AtT0);

    f_EUTRA_TestBody_Set (true);

    //@siclog "Step 1" siclog@

    //The SS transmits an RRCConnectionReconfiguration message to setup intra frequency measurement on Cell 1 & to

    //configure power preference indication.

    f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell1,

                                                    cs_RRCConnectionReconfiguration_8_2_2_6_2_Step1(v_RRC_TI,

                                                                                                  v_Frequency_IE_Cell2.UL_DL_Earfcn.dl_CarrierFreq,
                                                                                                  v_ChBandDependency_Cell2.AllowedMeasBandwidth,
                                                                                                  v_PowerPrefIndicationTimer));

    //@siclog "Step 2" siclog@

    /*WA# The UE transmits an RRCConnectionReconfigurationComplete message on Cell 1

    SRB.receive( car_SRB1_RrcPdu_IND( eutra_Cell1, cr_508_RRCConnectionReconfigurationComplete(v_RRC_TI)));*/
    //@siclog "Step 3" siclog@

    //Cause the UE to send UEAssistanceInformation message indicating "lowPowerConsumption". (Note1).

    f_UT_PowerPreferenceIndication(UT);

    //@siclog "Step 4" siclog@

    //UE transmits UEAssistanceInformation message with powerPrefIndication IE set to lowPowerConsumption

    SRB.receive (car_SRB1_RrcPdu_IND (eutra_Cell1, cr_508_UeAssistanceInformation (lowPowerConsumption)));

    //@siclog "Step 5" siclog@

    //Cause the UE to send UEAssistanceInformation message indicating "normal" (Note2).

    f_UT_PowerPreferenceIndication(UT,'0'B);

    //@siclog "Step 6" siclog@

    //UE transmits UEAssistanceInformation message with powerPrefIndication IE set to normal

    //and starts timer T340 with the timer value set to the powerPrefIndicationTimer.

    SRB.receive (car_SRB1_RrcPdu_IND (eutra_Cell1, cr_508_UeAssistanceInformation (normal)));

    t_T340.start(v_T340);

    //@siclog "Step 7" siclog@

    //The SS changes Cell 1 and Cell 2 parameters according to the row "T1" in table 8.2.2.6.2.3.2-1

    f_EUTRA_SetCellPowerList (v_CellPowerList_AtT1);

    //@siclog "Step 8" siclog@

    //The UE transmits a MeasurementReport message on Cell 1 to report event A3 with the measured RSRP,

    //RSRQ value for Cell 2
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1,

                                    cr_MeasurementReport_8_2_2_6_2_Step8(v_PhysCellId_Cell2)));
    //Configure SS to not send response on RACH preamble reception on Cell 2
    f_EUTRA_SS_ConfigRachProcedure ( eutra_Cell2,
                                     omit,

                                     cs_RachProcedureConfig_NoResponse ( f_EUTRA_CellInfo_GetDL_ChBandwidth ( eutra_Cell2 )));
    t_T304.start(v_T304);

    //@siclog "Step 9" siclog@

    //The SS transmits an RRCConnectionReconfiguration message on Cell 1 to order the UE

    //to perform intra frequency handover to Cell 2
    //WA#
    v_RRCConnectionReconfiguration := f_EUTRA_RRCConnectionReconfiguration_IntraLTE_HO_InterCell(eutra_Cell1,

                                                                                                 eutra_Cell2,

                                                                                                 tsc_C_RNTI_Def2,

                                                                                                 omit,

                                                                                                 n4,

                                                                                                 omit,

                                                                                                 false);

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell1, cs_TimingInfo_Now, v_RRCConnectionReconfiguration));                                                                                             

    /*

    EXCEPTION: The steps 1 and 2 below are repeated for the duration of T304.

     1- The UE attempts to perform the intra frequency handover using MAC Random Access Preamble on Cell 2.

     2- The SS does not respond.

    */

    //WA# Reset SRBs and DRBs of source cell (cell 1)

    f_EUTRA_SS_SRBs_DRBs_Reset_StopULGrant_noPUCCHSync(eutra_Cell1);
    t_T304.timeout;

    //@siclog "Step 10" siclog@

    //The SS changes Cell 1 and Cell 3 parameters according to the row "T2" in table 8.2.2.6.2.3.2-1

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT2);

    //Calculate ShortMAC-I to be received in RRCConnectionReestablishment Request from UE on Step11

    v_ShortMAC_I := f_Calculate_ShortMAC ( eutra_Cell1, eutra_Cell1 ); //WA#: Change the second parameter to cell 1 to correctly calculate ShortMAC
    //@siclog "Step 11" siclog@

    //The UE releases the configured powerPrefIndicationConfig, stops timer T340

    //and transmit an RRCConnectionReestablishmentRequest message on Cell 1

    t_T340.stop;

    SRB.receive (car_SRB0_RrcPdu_IND (eutra_Cell1,

                                      cr_508_RRCConnectionReestablishmentRequest (tsc_C_RNTI_Def,

                                                                                  v_PhysCellId_Cell1,

                                                                                  cr_ReestablishmentCause_HandoverFailure,

                                                                                  v_ShortMAC_I)));

    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest(eutra_Cell1, f_EUTRA_Security_Get(), v_Ncc0); //WA#
    f_EUTRA_Security_Set(v_Auth_Params); //WA#
    //@siclog "Step 12" siclog@

    //The SS transmits an RRCConnectionReestablishment message to resume SRB1 operation and re-activate security on Cell 1

    f_EUTRA_RRC_RRCConnectionReestablishment_Def (eutra_Cell1); //WA#
    //@siclog "Step 13" siclog@

    //The UE transmits an RRCConnectionReestablishmentComplete message on Cell 1

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_508_RRCConnectionReestablishmentComplete(v_RRC_TI)));

    //@siclog "Step 14" siclog@

    //Cause the UE to send UEAssistanceInformation message indicating "lowPowerConsumption". (Note1).

    f_UT_PowerPreferenceIndication(UT,-,true);

    //@siclog "Step 15" siclog@

    //Check: Does the UE transmit UEAssistanceInformation message within the next 20s? (Note 3)
    fl_WaitForUeAssistanceInformation(-, "Test Case 8.2.2.6.2 Step 15");
    //Table 8.2.2.6.2.3.3-12: OtherConfig-r9

    v_PowerPrefIndicationTimer := s10;

    //@siclog "Step 16" siclog@

    //The SS transmits an RRCConnectionReconfiguration message to resume existing radio bearer on Cell 1

    //and to configure power preference indication

    f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell1,

                                                    cs_RRCConnReconfig_Resume_SRB2_DRBs_ConfigurePowerPreferenceIndication(v_RRC_TI,

                                                                                        v_EUTRA_FDD_TDD_CellInfo_Cell1.cqi_ReportConfig,

                                                                                        v_EUTRA_FDD_TDD_CellInfo_Cell1.soundingRS_UL_Config,

                                                                                        v_AntennaInfo_Cell1,

                                                                                        v_SR_PUCCH_ResourceIndex_Cell1,

                                                                                        v_SR_ConfigIndex_Cell1,

                                                                                        omit,

                                                                                        omit,

                                                                                        v_PowerPrefIndicationTimer));

    //@siclog "Step 17" siclog@

    /*WA# The UE transmits an RRCConnectionReconfigurationComplete message on Cell 1

    SRB.receive( car_SRB1_RrcPdu_IND( eutra_Cell1, cr_508_RRCConnectionReconfigurationComplete(v_RRC_TI)));*/
    //@siclog "Step 18" siclog@

    //Cause the UE to send UEAssistanceInformation message indicating "lowPowerConsumption". (Note1).

    f_UT_PowerPreferenceIndication(UT,-,true);

    //@siclog "Step 19" siclog@

    //Check: Does the UE transmit UEAssistanceInformation message with powerPrefIndication IE

    //set to lowPowerConsumption? (Note 4)

    v_T340 := f_EUTRA_SetTimerToleranceMax ( eutra_Cell1, rrcTimer, 10.0 ); //s10

    t_T340.start(v_T340);

    alt{

        [] SRB.receive (car_SRB1_RrcPdu_IND (eutra_Cell1, cr_508_UeAssistanceInformation (lowPowerConsumption)))

        {

            t_T340.stop;

            f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 8.2.2.6.2 Step 19");

        }

        [] t_T340.timeout

        {

            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 8.2.2.6.2 Step 19");

        }

    }

    f_EUTRA_TestBody_Set (false);

    //Switch/power off UE

    f_EUTRA_Postamble (eutra_Cell1, E2_CONNECTED);

  }//end of TC8.2.2.6.2
... 

Change 2 – Correction to function ‘f_GetTestcaseAttrib_DlCcchMsgInSeparateMacPdu’ 
	Function name
	f_GetTestcaseAttrib_DlCcchMsgInSeparateMacPdu ()

	Reason for change
	TC 8.2.2.6.2 should be added to the list to stop the TTCN from sending pre-configured RRC Connection Setup message. This is because in this TC a HO failure is initiated at Step 9 and when the UE tries to re-establish connection on EUTRA cell 1, this results in the pre-configured RRC Connection Setup message sent along with contention resolution in response to the RRC Connection Re-establisment Request message.

	Summary of change
	Added TC 8.2.2.6.2 to the list.

	TTCN module
	\LTE_A_R10_R11\common\TestcaseProperties.ttcn

	MCC160 Comment
	


Before change

...

function f_GetTestcaseAttrib_DlCcchMsgInSeparateMacPdu(charstring p_Testcase) return boolean

  {

    select (p_Testcase) {

      case ("TC_17_4_7") { return true; }

      case ("TC_7_3_4_3") { return true; }

      case ("TC_8_2_4_23_1") { return true; }

      case ("TC_8_2_4_23_2") { return true; }

      case ("TC_8_2_4_23_3") { return true; }

      case ("TC_8_6_2_12") { return true; }

      case ("TC_8_6_2_5") { return true; }

      case ("TC_8_6_4_1") { return true; }

      case ("TC_8_6_4_10") { return true; }

      case ("TC_8_6_4_2") { return true; }

      case ("TC_8_6_4_3") { return true; }

      case ("TC_8_6_4_4") { return true; }

      case ("TC_8_6_4_5") { return true; }

      case ("TC_8_6_4_7") { return true; }

      case ("TC_8_6_4_9") { return true; }

      case ("TC_8_6_6_1") { return true; }

      case ("TC_8_6_6_2") { return true; }

      case ("TC_8_6_6_4") { return true; }

      case ("TC_8_6_6_5") { return true; }

      case ("TC_8_6_6_6") { return true; }

      case ("TC_8_6_6_7") { return true; }

      case ("TC_8_6_8_1") { return true; }

      case ("TC_8_6_8_2") { return true; }

      case ("TC_8_6_8_3") { return true; }

      case ("TC_8_6_8_5") { return true; }

      case ("TC_8_6_8_6") { return true; }

    }

    return false;

  }
...

After change

...

function f_GetTestcaseAttrib_DlCcchMsgInSeparateMacPdu(charstring p_Testcase) return boolean

  {

    select (p_Testcase) {

      case ("TC_17_4_7") { return true; }

      case ("TC_7_3_4_3") { return true; }

      case ("TC_8_2_2_6_2") { return true; } //WA#
      case ("TC_8_2_4_23_1") { return true; }

      case ("TC_8_2_4_23_2") { return true; }

      case ("TC_8_2_4_23_3") { return true; }

      case ("TC_8_6_2_12") { return true; }

      case ("TC_8_6_2_5") { return true; }

      case ("TC_8_6_4_1") { return true; }

      case ("TC_8_6_4_10") { return true; }

      case ("TC_8_6_4_2") { return true; }

      case ("TC_8_6_4_3") { return true; }

      case ("TC_8_6_4_4") { return true; }

      case ("TC_8_6_4_5") { return true; }

      case ("TC_8_6_4_7") { return true; }

      case ("TC_8_6_4_9") { return true; }

      case ("TC_8_6_6_1") { return true; }

      case ("TC_8_6_6_2") { return true; }

      case ("TC_8_6_6_4") { return true; }

      case ("TC_8_6_6_5") { return true; }

      case ("TC_8_6_6_6") { return true; }

      case ("TC_8_6_6_7") { return true; }

      case ("TC_8_6_8_1") { return true; }

      case ("TC_8_6_8_2") { return true; }

      case ("TC_8_6_8_3") { return true; }

      case ("TC_8_6_8_5") { return true; }

      case ("TC_8_6_8_6") { return true; }

    }

    return false;

  }
... 

5. Execution Log Files

5.1 MTK Jade Handset
The MTK Jade UE passed this EUTRA eDDA  test case on R&S CMW500 Multi-RAT Protocol Tester in eFDD band 3. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log :
Rohde-Schwarz\tc_8_2_2_6_2_MTK.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
In the log provided, masking is applied to prevent any disclosure of confidential information.
In the log file (in .txt ) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
References

	[1]
	R5s160038:    Supporting information for agreement of TC 8.2.2.6.2 in LTE-FDD mode.
                          This archive comprises:

                        -  text format execution log files 

	
	


